última atualização em 08/11/2008)

UNIVERSIDADE FEDERAL DO RIO GRANDE DO SUL

INSTITUTO DE MATEMÁTICA
DEPARTAMENTO DE MATEMÁTICA PURA E APLICADA
Disciplina: MAT01168 -Matemática Aplicada -Semestre Letivo 2008/2
Professoras: Elisabeta Gallicchio e Irene Strauch
TERCEIRA ÁREA

SÉRIES DE FOURIER UTILIZANDO MAPLE

Exercício: Uma tensão senoidal 10 sen(2t) atravessa um retificador de meia onda que absorve a porção negativa da onda. Desenvolva a função periódica resultante:

u(t)=0 quando -T/2<t<0
u(t)=10sen(2t) quando 0<t<T/2,

de período T=2π/w em uma série de Fourier. Trace os gráficos de u(t) e de sua série Truncada para 2 termos, 5 termos, 15 termos.

SOLUÇÃO

A função u(t) é representada pela série: $u(t) = \frac{a_0}{2} + \sum_{n=1}^{\infty}\left[a_n \cos\left(\frac{n\pi t}{L}\right) + b_n \sin\left(\frac{n\pi t}{L}\right)\right]$

onde os coeficientes an e bn são dados por

$$a_n = \frac{1}{L}\left(\int_{-L}^{L} u(t)\cos\left(\frac{n\pi t}{L}\right)\mathrm{d}t\right) \qquad \frac{a_0}{2} = \frac{1}{2L}\left(\int_{-L}^{L} u(t)\,\mathrm{d}t\right)$$

$$b_n = \frac{1}{L}\left(\int_{-L}^{L} u(t)\sin\left(\frac{n\pi t}{L}\right)\mathrm{d}t\right)$$

Observa-se que a freqüência é w=2 rad/s e o período fundamental é T=2*Pi/w. Também, verifica-se que este é um problema do CASO I, a função está definida em [-L,L]. Então, acessando os módulos gráficos e introduzindo os dados, tem-se

```
> restart:with(plots):with(plottools):assume(n,integer):interface
  (showassumed=0):#Para não exibir n~
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined
```

```
> w:=2; T:=2*Pi/w;
```

$$w := 2 \qquad \textbf{(1.1)}$$

$$T := \pi$$

```
> L := T/2;
```

$$L := \frac{\pi}{2} \qquad (1.2)$$

```
> u(t):=piecewise(t>-L and t<0 ,0,0<t and t<L,10*sin(2*t));
```

$$u(t) := \begin{cases} 0 & -\frac{\pi}{2} - t < 0 \text{ and } t < 0 \\ 10\sin(2t) & -t < 0 \text{ and } t - \frac{\pi}{2} < 0 \end{cases} \qquad (1.3)$$

Traçar o gráfico da função u(t):

```
> plot(u(t),t=-L..L, thickness=3,discont=true,titlefont=[COURIER,
  DEFAULT,16],labels=[t, "u(t)"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16], title="Gráfico da Tensão
  Senoidal");
```



Determinar os coeficientes a_n da série de Fourier, para n =1,2,3,..

```
> a[n]:=int(u(t)*cos(n*Pi/L*t),t=-L..L)/L;
```

$$a_{n\sim} := -\frac{10\left((-1)^{n\sim} + 1\right)}{\left(n\sim^2 - 1\right)\pi} \qquad (1.4)$$

Note que para n=1,tem-se divisão por zero nos a_n

Calcular o coeficiente a_0

```
> a[0]:= simplify(int(u(t),t=-L..L)/L);
```

$$a_0 := \frac{20}{\pi} \qquad (1.5)$$

A seguir, determinar os coeficientes b_n, para n=1,2,3,...

```
> b[n]:=int(u(t)*sin(n*Pi/L*t),t=-L..L)/L;
```

$$b_{n\sim} := 0 \qquad (1.6)$$

Observe que a resposta é b_n=0 para n=1,2,3,... Mas não está correta, como mostrado a seguir:

```
> a[1]:=int(u(t)*cos(Pi/L*t),t=-L..L)/L;
```

$$a_1 := 0 \qquad (1.7)$$

```
> b[1]:=int(u(t)*sin(1*Pi/L*t),t=-L..L)/L;
```

$$b_1 := 5 \tag{1.8}$$

Aproximar a função u(t) com a série truncada n=2 termos, com b_1=0

```
> u_ap:=a[0]/2 + sum(a[n]*cos(n*Pi/L*t)+b[n]*sin(n*Pi/L*t),n=2..2);
```

$$u\_ap := \frac{10}{\pi} - \frac{20}{3}\,\frac{\cos(4\,t)}{\pi} \tag{1.9}$$

Traçar os gráficos da função dada e da sua aproximação em série de Fourier truncada. Veja que não está certo. Falta o termo $b_1$sin(πt/L)

```
> plot({u(t),u_ap},t=-L..L,discont=true,titlefont=[COURIER,DEFAULT,
  12],labels=[t, "u(t)"],labelfont=[COURIER,DEFAULT,16],axesfont=
  [COURIER,DEFAULT,16],title=`u(t) e a Série de Fourier sem o termo
  b[1]sin(Pit/L)`, thickness=2, color=[aquamarine,blue]);
```



A seguir, a resposta correta, ou seja, com o termo $b_1$sin(πt/L)

```
> u_ap1:=a[0]/2+ b[1]*sin(Pi/L*t) + sum(a[n]*cos(n*Pi/L*t)+b[n]*sin
  (n*Pi/L*t),n=2..2);
```

$$u\_ap1 := \frac{10}{\pi} + 5\sin(2\,t) - \frac{20}{3}\,\frac{\cos(4\,t)}{\pi} \tag{1.10}$$

```
> plot({u(t),u_ap1},t=-L..L,discont=true,titlefont=[COURIER,
  DEFAULT,12],labels=[t, "u(t)"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=`u(t) e a Série de Fourier
  com o termo b[1]sin(Pit/L)`, thickness=2, color=[aquamarine,blue]
  );
```



Aproximar a função u(t) com a série truncada até n=5 .

```
> u_ap5:=a[0]/2+ b[1]*sin(Pi/L*t)+sum(a[n]*cos(n*Pi/L*t),n=2..5);
```

$$u\_ap5 := \frac{10}{\pi} + 5\sin(2\,t) - \frac{20}{3}\,\frac{\cos(4\,t)}{\pi} - \frac{4}{3}\,\frac{\cos(8\,t)}{\pi} \tag{1.11}$$

```
> plot({u(t),u_ap5},t=-L..L,discont=true,thickness=2,color=[red,
  black],titlefont=[COURIER,DEFAULT,12],labels=[t, "u(t)"],
  labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,16],
  title=` u(t) e a Série de Fourier truncada em n=5`);
```



Aproximar a função u(t) com a série truncada em n=15. Observa-se que a aproximação é perfeita, aparenta ser apenas um gráfico.

```
> u_ap15:=a[0]/2+ b[1]*sin(Pi/L*t)+sum(a[n]*cos(n*Pi/L*t),n=2..15);
```

$$u\_ap15 := \frac{10}{\pi} + 5\sin(2\,t) - \frac{20}{3}\,\frac{\cos(4\,t)}{\pi} - \frac{4}{3}\,\frac{\cos(8\,t)}{\pi} - \frac{4}{7}\,\frac{\cos(12\,t)}{\pi} - \frac{20}{63}\,\frac{\cos(16\,t)}{\pi} - \frac{20}{99}\,\frac{\cos(20\,t)}{\pi} - \frac{20}{143}\,\frac{\cos(24\,t)}{\pi} - \frac{4}{39}\,\frac{\cos(28\,t)}{\pi} \tag{1.12}$$

```
> plot({u(t),u_ap15},t=-L..L,discont=true,thickness=2, color=[red,
  black],titlefont=[COURIER,DEFAULT,12],labels=[t, "u(t)"],
  labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,16],
  title=`u(t) e a Série de Fourier truncada em n=15`);
```


u(t) e a S�rie de Fourier truncada em n=15
10
8
6
u(t)
4
2
-1.5
-1
-0.5
0
0.5
1
1.5
t

===